/*
  SD card test

 This example shows how use the utility libraries on which the'
 SD library is based in order to get info about your SD card.
 Very useful for testing a card when you're not sure whether its working or not.

 The circuit:
 * SD card attached to SPI bus as follows:
 ** UNO:  MOSI - pin 11, MISO - pin 12, CLK - pin 13, CS - pin 4 (CS pin can be
 changed) and pin #10 (SS) must be an output
 ** Mega:  MOSI - pin 51, MISO - pin 50, CLK - pin 52, CS - pin 4 (CS pin can be
 changed) and pin #52 (SS) must be an output
 ** Leonardo: Connect to hardware SPI via the ICSP header
    Pin 4 used here for consistency with other Arduino examples


 created  28 Mar 2011  by Limor Fried
 modified 9 Apr 2012   by Tom Igoe
 modified 16 Aug 2017  by Lambor Fang

 */
// include the SD library:
#include <SD.h>

// set up variables using the SD utility library functions:
Sd2Card  card;
SdVolume volume;
SdFile   root;

// change this to match your SD shield or module;
// Arduino Ethernet shield: pin 4
// Adafruit SD shields and modules: pin 10
// Sparkfun SD shield: pin 8
const int chipSelect = 43;

void setup() {
    // Open serial communications and wait for port to open:
    // SerialUSB.begin(115200);
    //  while (!Serial) {
    //   ; // wait for serial port to connect. Needed for Leonardo only
    // }

    SerialUSB.print( "\nInitializing SD card..." );
    // On the Ethernet Shield, CS is pin 4. It's set as an output by default.
    // Note that even if it's not used as the CS pin, the hardware SS pin
    // (10 on most Arduino boards, 53 on the Mega) must be left as an output
    // or the SD library functions will not work.
    pinMode( SS, OUTPUT );

    // we'll use the initialization code from the utility libraries
    // since we're just testing if the card is working!
    while ( !card.init( SPI_HALF_SPEED, chipSelect ) ) {
        SerialUSB.println( "initialization failed. Things to check:" );
        SerialUSB.println( "* is a card is inserted?" );
        SerialUSB.println( "* Is your wiring correct?" );
        SerialUSB.println( "* did you change the chipSelect pin to match your shield or module?" );
    }

    // print the type of card
    SerialUSB.print( "\nCard type: " );
    switch ( card.type() ) {
    case SD_CARD_TYPE_SD1:
        SerialUSB.println( "SD1" );
        break;
    case SD_CARD_TYPE_SD2:
        SerialUSB.println( "SD2" );
        break;
    case SD_CARD_TYPE_SDHC:
        SerialUSB.println( "SDHC" );
        break;
    default:
        SerialUSB.println( "Unknown" );
    }

    // Now we will try to open the 'volume'/'partition' - it should be FAT16 or
    // FAT32
    if ( !volume.init( card ) ) {
        SerialUSB.println( "Could not find FAT16/FAT32 partition.\nMake sure you've "
                           "formatted the card" );
        return;
    }

    // print the type and size of the first FAT-type volume
    uint32_t volumesize;
    SerialUSB.print( "\nVolume type is FAT" );
    SerialUSB.println( volume.fatType(), DEC );
    SerialUSB.println();

    volumesize = volume.blocksPerCluster();  // clusters are collections of blocks
    volumesize *= volume.clusterCount();     // we'll have a lot of clusters
    volumesize *= 512;                       // SD card blocks are always 512 bytes
    SerialUSB.print( "Volume size (bytes): " );
    SerialUSB.println( volumesize );
    SerialUSB.print( "Volume size (Kbytes): " );
    volumesize /= 1024;
    SerialUSB.println( volumesize );
    SerialUSB.print( "Volume size (Mbytes): " );
    volumesize /= 1024;
    SerialUSB.println( volumesize );

    SerialUSB.println( "\nFiles found on the card (name, date and size in bytes): " );
    root.openRoot( volume );

    // list all files in the card with date and size
    root.ls( LS_R | LS_DATE | LS_SIZE );
}

void loop( void ) {}
